

ivlpirf — Instrumental-variables local-projection impulse–response functions⁺

⁺This command is part of [StataNow](#).

[Description](#)
[Options](#)
[Acknowledgments](#)
[Quick start](#)
[Remarks and examples](#)
[References](#)
[Menu](#)
[Stored results](#)
[Also see](#)
[Syntax](#)
[Methods and formulas](#)

Description

`ivlpirf` uses instrumental variables (IV) to estimate impulse–response functions (IRFs) by local projections. Structural IRFs or cumulative structural IRFs for the response of one or more variables to an endogenous impulse variable may be estimated.

Quick start

IV local-projection IRFs for response variable `y`, endogenous impulse variable `x`, and instrument `z`

```
ivlpirf y, endogenous(x = z)
```

As above, but with 16 steps instead of the default 4

```
ivlpirf y, endogenous(x = z) step(16)
```

As above, but do not compute IRF for the endogenous impulse variable

```
ivlpirf y, endogenous(x = z) step(16) noendogirf
```

IV local-projection IRFs for response variables `y1`, `y2`, and `y3` with endogenous impulse variable `x` and instruments `z1` and `z2`

```
ivlpirf y1 y2 y3, endogenous(x = z1 z2)
```

As above, but request the cumulative impulse–response function instead of the default simple impulse–response function

```
ivlpirf y1 y2 y3, endogenous(x = z1 z2) cumulative
```

Menu

Statistics > Multivariate time series > Instrumental-variables local-projection IRFs

Syntax

```
ivlpirf depvarlist [if] [in], endogenous(depvariv = varlistiv) [options]
```

*depvar*_{iv} is an endogenous impulse variable.

*varlist*_{iv} is a list of instruments.

<i>options</i>	Description
Model	
* <u>endogenous</u> (<i>endospec</i>)	specify endogenous impulse variable and instruments
<u>lags</u> (<i>numlist</i>)	include specified lags of dependent and endogenous impulse variables; default is <code>lags(1 2)</code>
<u>step</u> (#)	set forecast horizon to # steps; default is <code>step(4)</code>
<u>exog</u> (<i>varlist</i> _{exo})	include exogenous variables as controls
<u>noconstant</u>	suppress constant terms in IV local projections
<u>noendogirf</u>	suppress IRF for the endogenous impulse variable
<u>cumulative</u>	report cumulative IRFs
SE/Robust	
<u>vce</u> (<i>vcetype</i>)	<i>vcetype</i> may be <code>robust</code> , <code>cluster <i>clustvar</i></code> , <code>conventional</code> , or <code>hac kernel #</code>
Reporting	
<u>level</u> (#)	set confidence level; default is <code>level(95)</code>
<u>display_options</u>	control columns and column formats and row spacing
<u>coeflegend</u>	display legend instead of statistics

* `endogenous()` is required. The full specification is `endogenous(depvariv = varlistiv)`.

You must `tsset` your data before using `ivlpirf`; see [TS] `tsset`.

*varlist*_{iv} and *varlist*_{exo} may contain factor variables; see [U] 11.4.3 **Factor variables**.

depvarlist, *depvar*_{iv}, *varlist*_{iv}, and *varlist*_{exo} may contain time-series operators; see [U] 11.4.4 **Time-series varlists**.

`by`, `collect`, `rolling`, and `statsby` are allowed; see [U] 11.1.10 **Prefix commands**.

`coeflegend` does not appear in the dialog box.

See [U] 20 **Estimation and postestimation commands** for more capabilities of estimation commands.

Options

Model

`endogenous(depvariv = varlistiv)` specifies the endogenous impulse variable *depvar*_{iv} and its instruments *varlist*_{iv}. `endogenous()` is required.

`lags(numlist)` specifies the lags of the dependent variables to be included in the model. The default is `lags(1 2)`. Lags may be skipped; for example, `lags(1 3)` would include lags 1 and 3 but not lag 2.

`step(#)` specifies the step (forecast) horizon; the default is four periods.

`exog(varlistexo)` specifies a list of exogenous variables to be included as controls in the IV local projections.

`noconstant` suppresses the constant terms in the IV local projections.

`noendogirf` suppresses the computation and reporting of the IRF for the endogenous impulse variable. By default, this IRF is calculated.

`cumulative` specifies that cumulative IRFs be calculated. The default is to calculate simple IRFs.

SE/Robust

`vce(vcetype)` specifies the type of standard error reported, which includes types that are robust to some kinds of misspecification (`robust`) and that allow for intragroup correlation (`cluster` *clustvar*); see [R] [vce_option](#).

`vce(conventional)` uses the unadjusted variance estimator from generalized method of moments (GMM); see [R] [gmm](#).

`vce(hac kernel #)` requests a heteroskedasticity- and autocorrelation-consistent (HAC) variance–covariance matrix using the specified kernel (see below) with # lags. The bandwidth of a kernel is equal to # + 1.

`vce(hac kernel opt [#])` requests a HAC variance–covariance matrix using the specified kernel, and the lag order is selected using Newey and West’s (1994) optimal lag-selection algorithm. # is an optional tuning parameter that affects the lag order selected; see the [discussion](#) in *Methods and formulas* in [R] [ivregress](#).

`vce(hac kernel)` requests a HAC weight matrix using the specified kernel and $N - 2$ lags, where N is the sample size.

There are three kernels available for HAC variance–covariance matrices, and you may request each one by using the name used by statisticians or the name perhaps more familiar to economists:

`bartlett` or `nwest` requests the Bartlett (Newey–West) kernel;

`parzen` or `gallant` requests the Parzen (Gallant 1987) kernel; and

`quadraticspectral` or `andrews` requests the quadratic spectral (Andrews 1991) kernel.

Reporting

`level(#)`, `nocnsreport`; see [R] [Estimation options](#).

`display_options`: `nocl`, `nopvalues`, `vsquish`, `cformat(%fmt)`, `pformat(%fmt)`, `sformat(%fmt)`, and `nolstretch`; see [R] [Estimation options](#).

The following option is available with `ivlpirf` but is not shown in the dialog box:

`coeflegend`; see [R] [Estimation options](#).

Remarks and examples

The local-projection method traces out the dynamic effect of an impulse variable on a response variable by using the sequence of projections

$$y_{t+h} = \beta_h x_t + \mathbf{w}'\boldsymbol{\gamma} + e_{t+h}$$

where y_{t+h} is the response variable at horizon h , x_t is the impulse variable, β_h is the impulse–response coefficient, variables in \mathbf{w} are controls with coefficients $\boldsymbol{\gamma}$, and e_{t+h} is an error term. All variables are measured over time t . The collection of coefficients $(\beta_0, \beta_1, \dots, \beta_H)$ traces out the response of y_t to an impulse in x_t . This is the local-projection estimator of the IRF. In many contexts, x_t is endogenous, rendering a causal interpretation of the IRF impossible. However, if an instrument z_t is available, then x_t may be instrumented with z_t , and the resulting IV local projections can be interpreted as a causal IRF.

Jordà and Taylor (forthcoming) provide an overview of the local-projection estimator, including the IV local-projection estimator used by `ivlpirf`, as well as applications of local projections to areas of applied economics.

▷ Example 1: IV local projections

We use data on U.S. industrial production growth (`ip_growth`), inflation rate (`inflation`), and the interest rate (`fedfunds`) to estimate the effects of an interest rate increase on economic activity and prices. At horizon h , we model the response of industrial production growth to change in interest rate as

$$\text{ip_growth}_{t+h} = \beta_h \Delta \text{fedfunds}_t + \mathbf{w}'\boldsymbol{\gamma} + e_{t+h}$$

where \mathbf{w} are controls, such as lags of `ip_growth` and change in `fedfunds`. We are concerned that the change in `fedfunds` is endogenous. We have available an instrument, `money_inst`, that captures monetary shocks. It is correlated with change in `fedfunds` but uncorrelated with any nonmonetary shocks. We use this variable as an instrument for change in `fedfunds`. In our `ivlpirf` command, we specify `d.fedfunds`, the first difference of `fedfunds`, as the endogenous impulse variable and the `money_inst` instrument in the `endogenous()` option.

```

. webuse usmacro3
(Federal Reserve Economic Data - St. Louis Fed, 2023-09-01)
. ivlpirf ip_growth, endogenous(d.fedfunds = money_inst)
Step 1:
Iteration 0: GMM criterion Q(b) = .22620934
Iteration 1: GMM criterion Q(b) = 4.665e-33
Iteration 2: GMM criterion Q(b) = 4.747e-34
Step 2:
Iteration 0: GMM criterion Q(b) = 2.524e-34
Iteration 1: GMM criterion Q(b) = 2.524e-34
note: model is exactly identified.
Instrumental-variables local-projection impulse responses
Sample: 1969m1 thru 2007m12                Number of obs = 468
( 1) [D.fedfunds]D.fedfunds = 1

```

	IRF coefficient	Robust std. err.	z	P> z	[95% conf. interval]	
ip_growth						
--.	.3902078	.2381631	1.64	0.101	-.0765833	.8569989
F1.	.3610046	.3089668	1.17	0.243	-.2445592	.9665685
F2.	.1629727	.2189905	0.74	0.457	-.2662408	.5921862
F3.	-.0997817	.1333441	-0.75	0.454	-.3611313	.1615678
F4.	-.1677662	.1170584	-1.43	0.152	-.3971965	.061664
fedfunds						
D1.	1 (constrained)					
FD.	1.476712	.5963349	2.48	0.013	.3079167	2.645506
F2D.	.3870066	.3807016	1.02	0.309	-.3591549	1.133168
F3D.	-.1729338	.2029923	-0.85	0.394	-.5707913	.2249238
F4D.	-.3460389	.1569646	-2.20	0.027	-.6536839	-.0383939

Note: Structural impulse–response functions are reported.

Impulse: D.fedfunds

Responses: ip_growth D.fedfunds

Instrument: money_inst

Controls: L.ip_growth L2.ip_growth LD.fedfunds L2D.fedfunds

The header of our `ivlpirf` output indicates that our model was fit using 468 observations of monthly data for January 1969 through December 2007.

The coefficient table displays the local-projection IRF coefficients. These coefficients are organized by response and step (horizon, in units of the time scale of the data). The `ip_growth` block of coefficients displays the response of industrial production growth to a shock to change in interest rates. The IRF coefficient of 0.39 labeled `--.` is the estimated instantaneous response to a shock to change in interest rate. The coefficient of 0.36 labeled `F1.` is the estimated response one month after the shock. The shocks labeled `F2.` and `F3.` are similarly the estimated responses after 2 and 3 months, respectively. The confidence intervals for these IRF coefficients all include 0; we don't find evidence of industrial production growth having a response to the shock in the first three months.

The `fedfunds` block of coefficients displays the response of change in `fedfunds` (`D.fedfunds`) to the shock in interest rate. The change interest rate itself rises by 1 unit in the impact period by construction, as we can see by the constrained value of 1 for the IRF coefficient. The scale of the data determines the units for interpreting the IRF coefficients. Because `fedfunds` is measured in percentage points, `d.fedfunds` is a percentage point change. The value 1 indicates a shock that raises the interest rate by 1%. Subsequent responses indicate a further change in the interest rate in the same direction, by 1.48%, in the first month after the shock. In the second period after the shock, there is further change in the interest rate in the same direction, by 0.39%. In subsequent periods,

the interest rate changes in the reverse direction, by -0.17% in the third period after a shock and finally by -0.35% in the fourth period.

Below the coefficient table, a note points out that the reported IRFs are structural IRFs; these provide a causal interpretation. We also see which variables are treated as the impulse, responses, instruments, and controls. Notice that, by default, two lags of each response (including the endogenous impulse variable) are treated as controls, but you may specify other controls. The `lags()` option modifies which lags of responses are to be used as controls, and the `exog()` option specifies additional exogenous variables to use as controls.

◀

► Example 2: Cumulative local projections

Consider again the model in example 1. This time, we compute cumulative IV local-projection IRFs. When the response is measured in growth rates, the cumulative IRF can be interpreted as the path of the level of the response variable, not the path of the growth rate, after an impulse. We are interested in the response of industrial production, so we use the `cumulative` option to compute the cumulative IRF for `ip_growth`.

```
. ivlpirf ip_growth, endogenous(d.fedfunds = money_inst) cumulative
Step 1:
Iteration 0: GMM criterion Q(b) = 1.814879
Iteration 1: GMM criterion Q(b) = 1.430e-32
Iteration 2: GMM criterion Q(b) = 4.570e-33
Step 2:
Iteration 0: GMM criterion Q(b) = 1.678e-32
Iteration 1: GMM criterion Q(b) = 1.678e-32
note: model is exactly identified.
Instrumental-variables local-projection impulse responses
Sample: 1969m1 thru 2007m12                Number of obs = 468
( 1) [D.fedfunds]D.fedfunds = 1
```

	CIRF coefficient	Robust std. err.	z	P> z	[95% conf. interval]	
ip_growth						
--.	.3902078	.2381631	1.64	0.101	-.0765833	.8569989
F1.	.7512125	.5259427	1.43	0.153	-.2796163	1.782041
F2.	.9141852	.7075975	1.29	0.196	-.4726805	2.301051
F3.	.8144034	.7943301	1.03	0.305	-.7424551	2.371262
F4.	.6466372	.7752842	0.83	0.404	-.872892	2.166166
fedfunds						
D1.	1 (constrained)					
FD.	2.476712	.5963349	4.15	0.000	1.307917	3.645506
F2D.	2.863718	.8591673	3.33	0.001	1.179781	4.547655
F3D.	2.690784	.93778	2.87	0.004	.8527694	4.528799
F4D.	2.344745	.8878207	2.64	0.008	.6046488	4.084842

Note: Cumulative structural impulse–response functions are reported.

Impulse: D.fedfunds

Responses: ip_growth D.fedfunds

Instrument: money_inst

Controls: L.ip_growth L2.ip_growth LD.fedfunds L2D.fedfunds


The cumulative response of industrial production growth on impact is the same as the simple response we obtained in example 1. The cumulative response in step 1, one month after impact, is the sum of the simple response on impact and the simple response in step 1 that were reported in example 1. This pattern continues.

These cumulative IRF coefficients now represent the response of industrial production rather than the response of industrial production growth. Thus, after one month, estimated response of industrial production to the shock is 0.75. After two months, the estimated response is 0.91. This estimated response then decreases in the next two periods.

Instead of showing the output in a table, we can graph the IRFs using `irf graph`. This time, we compute 36 steps of responses, corresponding to three years, by using the `step(36)` option. As controls, we include 12 lags (one year) of the `ip_growth` and `d.fedfunds` variables.

```
. ivlpirf ip_growth, endogenous(d.fedfunds = money_inst)
> cumulative step(36) lag(1/12)
(output omitted)
. irf set ivlpirf.irf, replace
(file ivlpirf.irf created)
(file ivlpirf.irf now active)
. irf create model1
(file ivlpirf.irf updated)
. irf graph csirf, impulse(D.fedfunds) yline(0) xlabel(0(12)36)
```

When we request cumulative IRFs in the `ivlpirf` command, we also need to request the cumulative IRFs, or more specifically cumulative structural IRFs, in the `irf graph` command. To do so, we specify the `csirf` statistic.



The graphs above display the cumulative IV local-projection IRF coefficients. Because both `d.fedfunds` and `ip_growth` are growth rates, both cumulative responses can be interpreted in levels of the variables. The left-hand panel shows the response of interest rate, and the right-hand panel shows the response of industrial production. On the left, we see that the interest rate rises after a shock to change in interest rate, with the response peaking a few periods after the shock but falling thereafter. On the right, we see that industrial production shows little movement on impact but then declines with a delay after the shock, reaching a trough about 24 periods (2 years) after the shock but then recovering slightly in the third year after the shock.

▷ Example 3: Simultaneous responses

We can compute multiple responses at once for the same endogenous impulse. We now add the inflation rate as a response variable and trace out the responses of inflation and industrial production growth. We thus have two local projections,

$$\text{ip_growth}_{t+h} = \beta_h \Delta \text{fedfunds}_t + \mathbf{w}'\gamma + e_{t+h}$$

and

$$\text{inflation}_{t+h} = \beta_h \Delta \text{fedfunds}_t + \mathbf{w}'\gamma + e_{t+h}$$

which we estimate simultaneously for all horizons.

```
. ivlpirf ip_growth inflation, endogenous(d.fedfunds = money_inst) cumulative
Step 1:
Iteration 0: GMM criterion Q(b) = 1.8287366
Iteration 1: GMM criterion Q(b) = 1.543e-32
Iteration 2: GMM criterion Q(b) = 5.190e-33
Step 2:
Iteration 0: GMM criterion Q(b) = 1.982e-32
Iteration 1: GMM criterion Q(b) = 1.982e-32
note: model is exactly identified.
Instrumental-variables local-projection impulse responses
Sample: 1969m1 thru 2007m12                                Number of obs = 468
( 1) [D.fedfunds]D.fedfunds = 1
```

	CIRF coefficient	Robust std. err.	z	P> z	[95% conf. interval]	
ip_growth						
--.	.3767377	.2067382	1.82	0.068	-.0284617	.7819371
F1.	.725611	.4473294	1.62	0.105	-.1511386	1.60236
F2.	.8729816	.5831643	1.50	0.134	-.2699994	2.015963
F3.	.7572531	.6219763	1.22	0.223	-.4617981	1.976304
F4.	.5751773	.567312	1.01	0.311	-.5367338	1.687088
inflation						
--.	.0725715	.0391172	1.86	0.064	-.0040968	.1492398
F1.	.1381907	.0682327	2.03	0.043	.0044571	.2719243
F2.	.2424316	.0933341	2.60	0.009	.0595001	.425363
F3.	.4166317	.1646479	2.53	0.011	.0939278	.7393355
F4.	.505858	.2142514	2.36	0.018	.085933	.925783
fedfunds						
D1.	1 (constrained)					
FD.	2.481322	.5854004	4.24	0.000	1.333959	3.628686
F2D.	2.867292	.8518579	3.37	0.001	1.197681	4.536903
F3D.	2.696978	.9320981	2.89	0.004	.8700996	4.523857
F4D.	2.354742	.8938336	2.63	0.008	.6028599	4.106623

Note: Cumulative structural impulse-response functions are reported.

Impulse: D.fedfunds

Responses: ip_growth inflation D.fedfunds


Instrument: money_inst

Controls: L.ip_growth L2.ip_growth L.inflation L2.inflation LD.fedfunds
L2D.fedfunds

There are three blocks in the output table, one each for the responses of industrial production growth, inflation rate, and the change in interest rate itself. By default, two lags of each response variable are included in the controls for each equation.

As before, we compute and graph three years of the cumulative impulse responses:

```
. ivlpirf ip_growth inflation, endogenous(d.fedfunds = money_inst)
> cumulative step(36) lag(1/12)
(output omitted)
. irf set ivlpirf.irf, replace
(file ivlpirf.irf created)
(file ivlpirf.irf now active)
. irf create model2
(file ivlpirf.irf updated)
. irf graph csirf, irf(model2) impulse(D.fedfunds) yline(0) xlabel(0(12)36)
```



Graphs by irfname, impulse variable, and response variable

The interest rate rises initially, then falls slowly (top-left panel). Industrial production declines, reaching a trough two years after the shock (bottom panel). New is the cumulative response of inflation rate (that is, the price level) in the top-right panel. Prices do not move much in the first two years after the shock but then finally show a slight decline in the third year.

Stored results

`ivlpirf` stores the following in `e()`:

Scalars

<code>e(N)</code>	number of observations
<code>e(N_gaps)</code>	number of gaps in sample
<code>e(k)</code>	number of parameters
<code>e(k_impulses)</code>	number of impulse variables
<code>e(k_responses)</code>	number of response variables
<code>e(k_instruments)</code>	number of instruments
<code>e(tmin)</code>	minimum time in sample
<code>e(tmax)</code>	maximum time in sample
<code>e(step)</code>	maximum step
<code>e(N_clust)</code>	number of clusters
<code>e(cumul)</code>	1 if option <code>cumulative</code> was specified; otherwise 0
<code>e(rank)</code>	rank of <code>e(V)</code>

Macros

<code>e(cmd)</code>	<code>ivlpirf</code>
<code>e(cmdline)</code>	command as typed
<code>e(depvar)</code>	names of dependent variables
<code>e(responses)</code>	names of response variables
<code>e(impulse)</code>	name of endogenous impulse variable
<code>e(instruments)</code>	names of instruments
<code>e(controls)</code>	names of control variables
<code>e(exog)</code>	names of exogenous variables
<code>e(title)</code>	title in estimation output
<code>e(tsfmt)</code>	format for the time variable
<code>e(tvar)</code>	time variable
<code>e(tmins)</code>	formatted minimum time
<code>e(tmaxs)</code>	formatted maximum time
<code>e(lags)</code>	lags used in controls
<code>e(cons)</code>	<code>noconstant</code> , if option specified
<code>e(clustvar)</code>	name of cluster variable
<code>e(vce)</code>	<code>vcetype</code> specified in <code>vce()</code>
<code>e(vcetype)</code>	title used to label Std. err.
<code>e(properties)</code>	<code>b V</code>
<code>e(estat_cmd)</code>	program used to implement <code>estat</code>
<code>e(predict)</code>	program used to implement <code>predict</code>

Matrices

<code>e(b)</code>	coefficient vector
<code>e(V)</code>	variance–covariance matrix of the estimators
<code>e(Cns)</code>	constraints matrix

Functions

<code>e(sample)</code>	marks estimation sample
------------------------	-------------------------

In addition to the above, the following is stored in `r()`:

Matrices

<code>r(table)</code>	matrix containing the coefficients with their standard errors, test statistics, p -values, and confidence intervals
-----------------------	---

Note that results stored in `r()` are updated when the command is replayed and will be replaced when any `r`-class command is run after the estimation command.

Methods and formulas

ivlpirf estimates local-projection IRFs by GMM. Coefficients for all responses at all horizons are computed simultaneously. Data are available for response variable y_t , endogenous impulse variable x_t , and instrument z_t , all measured over time indexed by $t = 1, 2, \dots, T$. The local projection for response y_t to an impulse x_t at horizon h is

$$y_{t+h} = \beta_h x_t + \mathbf{w}' \boldsymbol{\gamma} + e_{t+h}$$

where β_h is the coefficient of interest, \mathbf{w} is a vector of controls, and $\boldsymbol{\gamma}$ are coefficients on the controls. We have available an instrument z_t that is correlated with x_t and uncorrelated with e_t at all leads and lags. Note that this lead-lag exogeneity of the instrument is stronger than the usual requirement of lag exogeneity. To focus on the impulse–response coefficients, first partial out controls from the impulse variable, the response variable, and the instruments. These partialled-out variables are denoted y_t^\perp , x_t^\perp , and z_t^\perp , respectively. For the response variable y_t^\perp , at horizon h the moment condition is

$$E\{z_t^\perp (y_{t+h}^\perp - x_t^\perp \beta_h)\} = 0$$

All response coefficients for all variables are estimated simultaneously. This leads to a GMM problem,

$$\min_{\boldsymbol{\beta}} \left\{ \frac{1}{T} \sum \mathbf{z}_t^\perp (\mathbf{y}_H^\perp - x_t^\perp \boldsymbol{\beta}) \right\}' \mathbf{W} \left\{ \frac{1}{T} \sum \mathbf{z}_t^\perp (\mathbf{y}_H^\perp - x_t^\perp \boldsymbol{\beta}) \right\}$$

where \mathbf{y}_H is the $(H + 1) \times 1$ vector of responses at all desired horizons $(y_t, y_{t+1}, \dots, y_{t+H})$, $\boldsymbol{\beta}$ is the $(H + 1) \times 1$ vector of IRF coefficients, \mathbf{z} is a vector of instruments, and \mathbf{W} is a weight matrix. When there is more than one response variable, the vector \mathbf{y}_H is extended to include all horizons of all response variables, and $\boldsymbol{\beta}$ becomes the $(H + 1)k \times 1$ vector of IRF coefficients for k response variables at all horizons $0, 1, \dots, H$. For more details of GMM estimation, see [R] [ivregress](#) and [R] [gmm](#).

Cumulative responses are calculated by summing simple responses.

Acknowledgments

We thank Òscar Jordà and Alan Taylor of the University of California–Davis for helpful conversations.

References

- Andrews, D. W. K. 1991. Heteroskedasticity and autocorrelation consistent covariance matrix estimation. *Econometrica* 59: 817–858. <https://doi.org/10.2307/2938229>.
- Gallant, A. R. 1987. *Nonlinear Statistical Models*. New York: Wiley.
- Jordà, Ò., and A. M. Taylor. Forthcoming. Local projections. *Journal of Economic Literature*.
- Newey, W. K., and K. D. West. 1994. Automatic lag selection in covariance matrix estimation. *Review of Economic Studies* 61: 631–653. <https://doi.org/10.2307/2297912>.

Also see

[TS] **ivlpirf postestimation** — Postestimation tools for ivlpirf⁺

[TS] **lpirf** — Local-projection impulse–response functions

[TS] **tsset** — Declare data to be time-series data

[TS] **var ivsvar** — Instrumental-variables structural vector autoregressive models⁺

[R] **gmm** — Generalized method of moments estimation

[R] **ivregress** — Single-equation instrumental-variables regression

[U] **20 Estimation and postestimation commands**

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).